
1. Введение 
Язык IPL предназначен для описания задач цифровой обработки изображений. 

Фундаментальными типами языка являются вектор, матрица и древовидная двоич-
ная декомпозиция изображения. 
Программа на языке IPL может содержать: 

• глобальную секцию; 
• описания объектов фундаментальных типов; 
• описания преобразований, которые расширяют функциональность элемен-
тарных операций над объектами фундаментальных типов; 

• секцию целей. 
В глобальной секции определяются псевдонимы. Эта секция может содержать 

команду установки начального значения генератора случайных чисел. В секции це-
лей перечисляются подлежащие созданию объекты и растровые изображения. 

 

2. Грамматика языка 
Алфавит языка включает символы с ASCII-кодами от 33 до 255. Символы с ко-

дами от 0 до 32 являются пробельными. Символ # (знак номера) является знаком 
однострочного комментария. Если в лексему необходимо включить пробельный 
символ, то лексему следует заключить в двойные кавычки. 

 
Ниже дается формальное описание грамматики языка. Терминальные символы 

выделены полужирным шрифтом. 
 
IPL-программа: 
 IPL-программа | секция 
 
секция: 
 глобальная_секция | определение_объекта | \ 
 определение_преобразования | определение_функции | 
 секция_целей 
 
глобальная_секция: 
 global { оператор_глобальной_секции } 

 
оператор_глобальной_секции: 
 оператор_глобальной_секции | определение_псевдонима | \ 
 инициализация_генератора_случайных_чисел 

 



определение_псевдонима: 
 alias псевдоним значение_псевдонима 

 
псевдоним: 
 идентификатор 

 
идентификатор: 
 идентификатор | непробельный_символ 

 
значение_псевдонима: 
 идентификатор 

 
инициализация_генератора_случайных_чисел: 
 random_seed начальное_значение_ГСЧ 

 
начальное_значение_ГСЧ: 
 беззнаковый_параметр 

 
определение_объекта: 
 определение_вектора | определение_матрицы | \ 
 определение_декомпозиции 

 
определение_вектора: 
 vector имя_объекта { изменение_состояния | \ 
  векторная_операция } 

 
определение_матрицы: 
 matrix имя_объекта { изменение_состояния | \ 
  матричная_операция } 

 
определение_декомпозиции: 
 decomposition имя_объекта { изменение_состояния | \ 
  операция_декомпозиции } 

 
имя_объекта: 
 идентификатор 

 
изменение_состояния: \ 
 изменение_состояния | \ 
 state custom_width число_столбцов_по_умолчанию | \ 
 state custom_height число_строк_по_умолчанию | \ 
 state custom_size число_элементов_по_умолчанию | \ 
 state input_powers2 масштабирование_к_степени_двух | \ 
 state input_square масштабирование_к_равным_размерам | \ 
 state frequency_domain текущая_область_частотная | \ 
 state pad_value значение_заполнитель 

 
число_столбцов_по_умолчанию: 
число_строк_по_умолчанию: 
число_элементов_по_умолчанию: 
 беззнаковый_параметр 

 



масштабирование_к_степени_двух: 
масштабирование_к_равным_размерам: 
текущая_область_частотная: 
 логический_параметр 

 
значение_заполнитель: 
 вещественный_параметр 

 
векторная_операция: 
 векторная_операция | копирование_вектора | \ 
 сохранение_вектора | загрузка_вектора | \ 
 циклический_сдвиг | инверсия_порядка_элементов | \ 
 равномерно_распределенный_шум | \ 
 нормально_распределенный_шум | \ 
 установка_всех_элементов | линейная_интерполяция | \ 
 гистограмма | гистограмма_по_диапазону | \ 
 гистограмма_с_записью_координат_центров | \ 
 гистограмма_по_диапазону_с_записью_координат_центров | \ 
 дискретный_интеграл | \ 
 приведение_к_единичному_диапазону | \ 
 приведение_к_заданному_диапазону | \ 
 вызов_преобразования_вектора | \ 
 вызов_преобразования_пары_векторов | \ 
 печать_строки_на_консоли | печать_счетчика_времени 

 
копирование_вектора: 
 copy вектор_источник 

 
вектор_источник: 
 векторный_параметр 

 
сохранение_вектора: 
 save имя_файла 

 
загрузка_вектора: 
 load имя_файла 

 
циклический_сдвиг: 
 shift новое_смещение_начального_элемента 

 
новое_смещение_начального_элемента: 
 целочисленный_параметр 

 
инверсия_порядка_элементов: 
 mirror 

 
равномерно_распределенный_шум: 
 uniform нижняя_граница верхняя_граница 
 
нормально_распределенный_шум: 
 gaussian математическое_ожидание стандартное_отклонение 
 
нижняя_граница: 
верхняя_граница: 
математическое_ожидание: 
стандартное_отклонение: 
 вещественный_параметр 
 



установка_всех_элементов: 
 constant значение_заполнитель 

 
линейная_интерполяция: 
 linear нижняя_граница верхняя_граница 

 
гистограмма: 
 histogram матрица_источник 

 
матрица_источник: 
 матричный_параметр 

 
гистограмма_по_диапазону: 
 range_histogram матрица_источник \ 
  нижняя_граница верхняя_граница 

 
гистограмма_с_записью_координат_центров: 
 histogram_with_coords матрица_источник \ 
  вектор_приемник_координат 

 
вектор_приемник_координат: 
 векторный_параметр 

 
гистограмма_по_диапазону с_записью_координат_центров: 
 range_histogram_with_coords матрица_источник \ 
  нижняя_граница верхняя_граница \ 
  вектор_приемник_координат 

 
дискретный_интеграл: 
 integrate 

 
приведение_к_единичному_диапазону: 
 normalize 

 
приведение_к_заданному_диапазону: 
 range_normalize 

 
вызов_преобразования_вектора: 
 vector_transform имя_преобразования \ 
  ( void | список_параметров ) 

 
имя_преобразования: 
 идентификатор 

 
вызов_преобразования_пары_векторов: 
 vector_pair_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  второй_векторный_аргумент 

 
список_параметров: 
 вещественный_параметр | \ 
  вещественный_параметр , список_параметров 

 
второй_векторный_аргумент: 
 векторный_параметр 

 



печать_строки_на_консоли: 
 puts идентификатор 

 
печать_счетчика_времени: 
 run_time 

 
матричная_операция: 
 матричная_операция | копирование_матрицы | \ 
 сохранение_матрицы | загрузка_матрицы | \ 
 сохранение_в_формате_с_фиксированной_точкой | \ 
 загрузка_из_файла_в_формате_с_фиксированной_точкой | \ 
 загрузка_цветового_канала | \ 
 копирование_подматрицы | \ 
 установка_элемента | установка_всех_элементов | \ 
 установка_элементов_подматрицы | \ 
 циклический_сдвиг_элементов | \ 
 изменение_порядка_элементов | изменение_размера | \ 
 дополнение_заданным_значением | \ 
 дополнение_заданным_значением_до_степени_двух | \ 
 равномерно_распределенный_шум | \ 
 нормально_распределенный_шум | \ 
 линейная_интерполяция_по_строкам | \ 
 линейная_интерполяция_по_столбцам | \ 
 свертка_с_фильтром_размера_3х3 | \ 
 прямое_преобразование_фурье | \ 
 обратное_преобразование_фурье | \ 
 вычисление_оценки_спектра_мощности | \ 
 сложение_матриц | \ 
 сложение_всех_элементов_с_константой | \ 
 умножение_на_скаляр | поэлементное_умножение | \ 
 поэлементное_деление | \ 
 обнуление_строк | обнуление_столбцов | \ 
 обнуление_полосы | выделение_полосы | \ 
 приведение_данного_диапазона_контраста_к_единице | \ 
 приведение_полного_диапазона_контраста_к_единице | \ 
 приведение_полного _диапазона_контраста_к_данному | \ 
 нелинейная_коррекция_гаммы | \ 
 отображение_декомпозиции | \ 
 отображение_матрицы_уровней | \ 
 отображение_матрицы_уровней_оттенками_серого | \ 
 вызов_преобразования_матрицы | \ 
 вызов_преобразования_пары_матриц | \ 
 вызов_декартова_преобразования_фурье_образа | \ 
 вызов_декартова_преобразования_пары_фурье_образов | \ 
 вызов_полярного_преобразования_фурье_образа | \ 
 вызов_полярного_преобразования_пары_фурье_образов | \ 
 печать_строки_на_консоли | печать_счетчика_времени 

 
копирование_матрицы: 
 copy матрица_источник 

 
сохранение_матрицы: 
 save имя_файла 

 
загрузка_матрицы: 
 load имя_файла 

 
сохранение_в_формате_с_фиксированной_точкой: 
 save32 имя_растрового_файла_TARGA 

 



имя_растрового_файла_TARGA: 
 имя_файла 

 
загрузка_из_файла_в_формате_с_фиксированной_точкой: 
 load32 имя_растрового_файла_TARGA 

 
загрузка_цветового_канала: 
 input имя_файла_растрового_изображения имя_канала 
 
имя_файла_растрового_изображения: 
 строковая_константа 

 
имя_канала: 

red | green | blue | alpha | 
 cieX | cieY | cieZ 
 hue | saturation | value 
 
копирование_подматрицы: 
 get колонка_приемник строка_приемник \ 
  матрица_источник \ 
  колонка_источник строка_источник \ 
  последняя_колонка последняя_строка 

 
колонка_приемник: 
строка_приемник: 
 целочисленный_параметр 

 
колонка_источник: 
строка_источник: 
последняя_колонка: 
последняя_строка: 
 беззнаковый_параметр 

 
установка_элемента: 
 set_elem колонка строка значение_заполнитель 

 
колонка: 
строка: 
 беззнаковый_параметр 

 
установка_всех_элементов: 
 constant значение_заполнитель 

 
установка_элементов_подматрицы: 
 set_rect начальная_колонка начальная_строка \ 
  последняя_колонка последняя_строка \ 
  значение_заполнитель 

 
начальная_колонка: 
начальная_строка: 
 беззнаковый_параметр 

 
циклический_сдвиг_элементов: 
 shift новое_смещение_нулевой_колонки \ 
  новое_смещение_нулевой_строки 

 



новое_смещение_нулевой_колонки: 
новое_смещение_нулевой_строки: 
 целочисленный_параметр 

 
изменение_порядка_элементов: 
 mirror изменение_порядка_по_горизонтали \ 
  изменение_порядка_по_вертикали 

 
изменение_порядка_по_горизонтали: 
изменение_порядка_по_вертикали: 
 логический_параметр 

 
изменение_размера: 
 resize новое_число_колонок новое_число_строк 

 
новое_число_колонок: 
новое_число_строк: 
 беззнаковый_параметр 

 
дополнение_заданным_значением: 
 expand новое_число_колонок новое_число_строк 

 
дополнение_заданным_значением_до_степени_двух: 
 expand_powers2 

 
линейная_интерполяция_по_строкам: 
 col_linear значение_в_первой_колонке \ 
  значение_в_последней_колонке 

 
линейная_интерполяция_по_столбцам: 
 row_linear значение_в_первой_строке \ 
  значение_в_последней_строке 

 
значение_в_первой_колонке: 
значение_в_последней_колонке: 
значение_в_первой_строке: 
значение_в_последней_строке: 
 вещественный_параметр 

 
свертка_с_фильтром_размера_3х3: 
 convolve3x3 фильтр_3х3 

 
фильтр_3х3: 
 матричный_параметр 

 
прямое_преобразование_фурье: 
 fourier_forward 

 
обратное_преобразование_фурье: 
 fourier_inverse 

 



вычисление_оценки_спектра_мощности: 
 spectrum 

 
сложение_матриц: 
 mat_offset матричный_параметр 

 
сложение_всех_элементов_с_константой: 
 offset вещественный_параметр 

 
умножение_на_скаляр: 
 scale вещественный_параметр 

 
поэлементное_умножение: 
 mat_scale матричный_параметр 

 
поэлементное_деление: 
 mat_unscale матричный_параметр 

 
обнуление_строк: 
 zero_rows начальная_строка конечная_строка 

 
обнуление_столбцов: 
 zero_cols начальная_колонка конечная_колонка 

 
обнуление_полосы: 
 band_reject начальная_строка_и_колонка \ 
  конечная_строка_и_колонка 

 
выделение_полосы: 
 band_pass начальная_строка_и_колонка \ 
  конечная_строка_и_колонка 

 
начальная_строка_и_колонка: 
конечная_строка_и_колонка: 
 беззнаковый_параметр 

 
приведение_данного_диапазона_контраста_к_единице: 
 adjust_contrast значение_приводимое_к_нулю \ 
  значение_приводимое_к_единице 

 
значение_приводимое_к_нулю: 
значение_приводимое_к_единице: 
 вещественный_параметр 

 
приведение_полного_диапазона_контраста_к_единице: 
 normalize 

 
приведение_полного_диапазона_контраста_к_данному: 
 range_normalize новый_минимум новый_максимум 

 



новый_минимум новый_максимум 
 вещественный_параметр 

 
нелинейная_коррекция_гаммы: 
 gamma_correction корректирующая_кривая 

 
корректирующая_кривая: 
 векторный_параметр 

 
отображение_декомпозиции: 
 draw_decomposition параметр_декомпозиция 

 
отображение_матрицы_уровней: 
 draw_decomposition_scale параметр_декомпозиция \ 
  оттенки_серого 

 
оттенки_серого: 
 векторный_параметр 

 
отображение_матрицы_уровней_оттенками_серого: 
 draw_decomposition_linear параметр_декомпозиция \ 
  оттенок_для_начальной_глубины \ 
  оттенок_для_конечной_глубины 

 
оттенок_для_начальной_глубины: 
оттенок_для_конечной_глубины: 
 вещественный_параметр 

 
вызов_преобразования_матрицы: 
 matrix_transform имя_преобразования \ 
  ( void | список_параметров ) 

 
вызов_преобразования_пары_матриц: 
 matrix_pair_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  второй_матричный_аргумент 

 
второй_матричный_аргумент: 
 матричный_параметр 

 
вызов_декартова_преобразования_фурье_образа: 
 cartesian_transform имя_преобразования \ 
  ( void | список_параметров ) 

 
вызов_декартова_преобразования_пары_фурье_образов: 
 cartesian_pair_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  второй_матричный_аргумент 

 
вызов_полярного_преобразования_фурье_образа: 
 polar_transform имя_преобразования \ 
  ( void | список_параметров ) 

 



вызов_полярного_преобразования_пары_фурье_образов: 
 polar_pair_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  второй_матричный_аргумент 

 
операция_декомпозиции: 
 операция_декомпозиции | выполнение_декомпозиции | \ 
 сохранение_ декомпозиции | печать_строки_на_консоли | \ 
 печать_счетчика_времени 

 
выполнение_декомпозиции: 
 decompose матрица_источник порог \ 
  количество_разрядов_гистограмм_на_каждом_уровне 

 
порог: 
 вещественный_параметр 

 
количество_разрядов_гистограмм_на_каждом_уровне: 
 беззнаковый_параметр 

 
 
сохранение_декомпозиции: 
 save имя_файла сохранять_значения относительные_частоты 

 
сохранять_значения: 
относительные_частоты: 
 логический_параметр 

 
определение_преобразования: 
 преобразование_вектора | 
 преобразование_пары_векторов | 
 преобразование_матрицы | 
 преобразование_пары_матриц | 
 декартово_преобразование_фурье_образа | 
 декартово_преобразование_пары_фурье_образов | 
 полярное_преобразование_фурье_образа | 
 полярное_преобразование_пары_фурье_образов 

 
преобразование_вектора: 
 vector_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  тело_преобразования 

 
преобразование_пары_векторов: 
 vector_pair_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  тело_преобразования 

 
преобразование_матрицы: 
 matrix_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  тело_преобразования 

 



преобразование_пары_матриц: 
 matrix_pair_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  тело_преобразования 

 
декартово_преобразование_фурье_образа: 
 cartesian_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  тело_преобразования 

 
декартово_преобразование_пары_фурье_образов: 
 cartesian_pair_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  тело_преобразования 

 
полярное_преобразование_фурье_образа: 
 polar_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  тело_преобразования 

 
полярное_преобразование_пары_фурье_образов: 
 polar_pair_transform имя_преобразования \ 
  ( void | список_параметров ) \ 
  тело_преобразования 

 
тело_преобразования: 
 { оператор_преобразования } 

 
оператор_преобразования: 
 оператор_преобразования | \ 
 однократно_выполняемый_оператор | \ 
 оператор_выполняемый_для_каждого_элемента 

 
однократно_выполняемый_оператор: 
 step составной_оператор 

 
оператор_выполняемый_для_каждого_элемента: 
 pass составной_оператор 

 
составной_оператор: 
 оператор | { список_операторов } 

 
список_операторов: 
 | \ 
 список_операторов | \ 
 оператор 

 
оператор: 
 определение_переменной | \  
 оператор_присваивания | \  
 условный_оператор | \  
 оператор_возврата 

 
определение_переменной: 
 var список_переменных; 

 



список_переменных: 
 идентификатор | идентификатор , список_переменных 

 
оператор_присваивания: 
 идентификатор = выражение; 

 
условный_оператор: 
 if выражение составной_оператор | \ 
 if выражение составной_оператор else составной_оператор 

 
оператор_возврата: 
 return выражение; 

 
выражение: 
 вещественная_константа | \ 
 переменная | \ 
 вызов_функции | \ 
 выражение_в_обратной_польской_нотации 

 
выражение_в_обратной_польской_нотации: 
 ( список_действий ) 

 
список_действий: 
 операнд_или_операция | \ 
 операнд_или_операция , список_действий 

 
операнд_или_операция: 
 внутренняя_константа | \ 
 вещественная_константа | \ 
 внутренняя_переменная | \ 
 переменная | \ 
 вызов_функции | \ 
 операция 

 
внутренняя_константа: 
 Pi | E | MAXREAL | REAL_ZERO 

 
внутренняя_переменная: 
 $MAX_IDX | $IDX | $ELEM | \ 
 $ELEM1 | $ELEM2 | \ 
 $MAX_ROW | $MAX_COL | $ROW | $COL | $CELL | \ 
 $CELL1 | $CELL2 | \ 
 $NEG_FREQ | $RE | $IM | \ 
 $RE1 | $IM1 | $RE2 | $IM2 | \ 
 $MAG | $PHASE | \ 
 $MAG1 | $PHASE1 | $MAG2 | $PHASE2 

 
переменная: 
 идентификатор 

 
операция: 
 унарная_операция | бинарная_операция 

 



унарная_операция: 
 neg | abs | ceil | floor | sqr | sqrt | sin | csc | \ 
 cos | sec | tg | ctg | arcsin | arccsc | arccos | \ 
 arcsec | arctg | arcctg | exp | sh | ch | th | ln | \ 
 lg | log2 | arsh | arch | arth 

 
бинарная_операция: 
 + | - | * | / | mod | ^ | log | arctg2 | \ 
 < | > | == | != 

 
определение_функции: 
 function имя_функции тело_функции 

 
имя_функции: 
 идентификатор 
 
тело_функции: 
 составной_оператор 

 
секция_целей: 
 targets { цель } 

 
цель: 
 цель | целевой_объект | растровое_изображение 

 
целевой_объект: 
 имя_типа имя_объекта 

 
имя_типа: 
 vector | matrix | decomposition 

 
имя_объекта: 
 идентификатор 

 
растровое_изображение: 
 output имя_файла цветовое_пространство \ 
  канал_1 канал_2 канал_3 [канал_4] 

 
цветовое_пространство: 
 RGB | RGBA | HSV | CIE 
 
канал_1:  
 матричный_параметр 

 
канал_2: 
 матричный_параметр 

 
канал_3: 
 матричный_параметр 

 
канал_4: 
 матричный_параметр 

 



векторный_параметр: 
 идентификатор 
 
матричный_параметр: 
 идентификатор 
 
параметр_декомпозиция: 
 идентификатор 
 
вещественный_параметр: 
 вещественная_константа | вызов_функции 
 
логический_параметр: 
 true | false | вызов_функции 
 
целочисленный_параметр: 
 целочисленная_константа | вызов_функции 
 
беззнаковый_параметр: 
 беззнаковая_константа | вызов_функции 
 
вещественная_константа: 
 32-битное число с плавающей точкой 
 
целочисленная_константа: 
 число в диапазоне от -2147483648 до 2147483647 
 
беззнаковая_константа: 
 число в диапазоне от 0 до 4294967295 
 
вызов_функции: 
 имя_функции 
 
Интерпретатор является многопроходным. На первом проходе разбираются все 

глобальные секции. 
На втором проходе разбираются все секции целей. Если целевой объект или рас-

тровое изображение содержат ссылку на неопределенный объект фундаментального 
типа, то по тексте программы осуществляется поиск определения этого объекта (по-
этому порядок размещения определений не важен). 
Явная или неявная рекурсия в определениях невозможна. Например, разбор сле-

дующего фрагмент кода 
 
matrix a 
{ 
 expand 3 3 
 copy b 
} 
 
matrix b 
{ 
 state custom_width width_of a 
 state custom_height height_of a 
 constant 1 
} 
 



targets 
{ 
 matrix b 
} 

 
приводит к ошибке: 
 
ERROR in "test.ipl", line 4: circular reference:  (got "b"). 
(кольцевая ссылка в строке 4) 

 
Сообщение об аналогичной ошибке будет получено и в том случае, если в секции 

целей в качестве целевого объекта будет указана матрица a. 
 
Константы записываются по правилам, принятым для алгоритмических языков 

высокого уровня. Разделителем является десятичная точка (это не относится к опе-
рациям сохранения / восстановления векторов и матриц; символ-разделитель опре-
делен в файле istrings.h как макрос IPL_DOT_CHAR, в текущей версии – запятая). 
Внутренние переменные $MAX_IDX (максимально возможный индекс элемента) и 

$IDX (индекс текущего элемента) определены для преобразований вектора и пары 
векторов. Переменная $ELEM (значение текущего элемента) определена только для 
преобразования вектора; переменные $ELEM1 и $ELEM2 (значения текущих элемен-
тов первого – основного – и второго векторных аргументов) определены для преоб-
разования пары векторов. 
Переменные $MAX_ROW (максимально возможный номер строки), $MAX_COL (мак-

симально возможный номер колонки), $ROW и $COL (номера строки и колонки теку-
щего элемента) определены для всех преобразований матриц. 
Переменная $CELL (значение текущего элемента) допустима только для преобра-

зования матрицы; переменные $CELL1 и $CELL2 (значения текущих элементов пер-
вого – основного – и второго матричных аргументов) допустимы для преобразова-
ния пары матриц. 
Переменная $NEG_FREQ (установлена в 1 для отрицательных пространственных 

частот) определена для всех преобразований, выполняющихся над матрицами в об-
ласти пространственных частот. 
Переменные $RE и $IM (действительная и мнимая части значения текущего эле-

мента, находящегося в области пространственных частот) допустимы для преобра-
зования, определенного ключевым словом cartesian_transform; переменные $RE1, 
$IM1, $RE2 и $IM2 (действительные и мнимые части значений текущих элементов 
первого и второго матричных аргументов, находящихся в области пространствен-
ных частот) допустимы для преобразования, определенного ключевым словом car-
tesian_pair_transform. 
Переменные $MAG и $PHASE (модуль и аргумент значения текущего элемента, на-

ходящегося в области пространственных частот) допустимы для преобразования, 
определенного ключевым словом polar_transform; переменные $MAG1, $PHASE1, 
$MAG2 и $PHASE2 (модули и аргументы значений текущих элементов первого и вто-
рого матричных аргументов, находящихся в области пространственных частот) до-



пустимы для преобразования, определенного ключевым словом 
polar_pair_transform. 
Синтаксис языка является свободным. Например, пусть требуется сохранить в 

файл матрицу размера 5х5, каждый элемент которой равен . Соответствующая 
программа может выглядеть следующим образом: 

 
targets{matrix a}matrix a{state pad_value sqrt2 expand 5 5 save a}function 
sqrt2{return (2,sqrt);} 

 
Однако предпочтительнее записать ее текст в виде: 
 

matrix a 
{ 
 state pad_value sqrt2 
 expand 5 5 
 save a 
} 

 
function sqrt2 
{ 
 return (2,sqrt); 
} 

 
targets 
{ 
 matrix a 
} 

 
Функции не принимают аргументов. Единственной целью их введения является 

желание обойти одно из ограничений текущей версии – невозможность использова-
ния выражений в определениях объектов фундаментальных типов. 

3. Некоторые примеры использования языка IPL 
 
Программа, выводящая на консоль стандартное приветствие: 
 
matrix dummy 
{ 
 expand 1 1 
 puts "Hello, World !" 
} 
 
targets 
{ 
 matrix dummy 
} 

 
(dummy): Hello, World ! 
IPL run completed: success. (alloc: 1035 bytes; leak: 0 bytes) 

 
Выглядит несколько искусственно – следует учесть, что язык не создавался для 

упражнений подобного рода. 
 



Создание бесшовной текстуры «плазма» (каналом цветового тона является низко-
частотная составляющая гауссова шума): 

 
global 
{ 
 random_seed 2 
 alias size 512 
 alias mean 0.6 
 alias stddev 2 
 alias max_pass 3 
 alias file_name "out.bmp" 
 alias clamp_min 0.1  # from orange ... 
 alias clamp_max 0.8  # to violet 
} 
 
matrix unit 
{ 
 state pad_value 1 
 expand size size 
} 
 
matrix hue 
{ 
 expand width_of unit height_of unit 
 gaussian mean stddev 
 run_time 
 fourier_forward 
 run_time 
 band_pass 0 max_pass 
 fourier_inverse 
 normalize # current range is [0..1] 
 matrix_transform clamp(clamp_min,clamp_max) 
} 
 
matrix_transform clamp(min_val,max_val) 
{ 
 pass 
 { 
  if($CELL,min_val,<) 
   $CELL = min_val; 
  else if($CELL,max_val,>) 
   $CELL = max_val; 
 } 
} 
 
targets 
{ 
 output file_name HSV hue unit unit 
} 
 

(hue): time since start: 0.27 
(hue): time since start: 0.42 
IPL run completed: success. (alloc: 2118204 bytes; leak: 0 bytes) 

 



Время выполнения двумерного преобразования Фурье для 
матрицы размеров 512х512 составляет 150 мс (компьютер с 
ЦПУ PII-300 МГц). Результат работы программы представлен 
на рис. 1. 

 
Формирование шкалы цветовых тонов: 
 
global 
{ 
 alias bmp_width 1280 
 alias bmp_height 100 
} 
 
matrix const_mat 
{ 
 state custom_width bmp_width 
 state custom_height bmp_height 
 constant 1 
} 
 
matrix h_grad 
{ 
 state custom_width bmp_width 
 state custom_height bmp_height 
 row_linear 0 1 
} 
 
matrix v_grad 
{ 
 state custom_width bmp_width 
 state custom_height bmp_height 
 col_linear 1 0 
} 
 
targets 
{ 
 output h_scale.bmp HSV h_grad v_grad const_mat 
} 
 

IPL run completed: success. (alloc: 1540874 bytes; leak: 0 bytes) 
 
Результат представлен на рис. 2. 
 

 
Рис. 2 

 
Приведенная ниже программа помещает канал яркости растрового изображения 

in.bmp (рис. 3) в матрицу src и затем вычисляет гистограмму яркости: 
 

 
Рис. 1 



global 
{ 
 alias divs 100 
} 
 
matrix src 
{ 
 input "in.bmp" value 
} 
 
vector out 
{ 
 state custom_size divs 
 range_histogram src 0 1 
 save out.txt 
} 
 
targets 
{ 
 vector out 
} 

 
IPL run completed: success. (alloc: 8408892 bytes; leak: 0 bytes) 

 
Результат построения гистограммы по данным из файла out.txt представлен на 

рис. 4. 
 

 

 
Рис. 3. 

Рис. 4 

 


